# Maintainer: Orhun Parmaksız <orhun@archlinux.org>

pkgname=shy
pkgver=0.1.10
pkgrel=3
pkgdesc="Bare-bones SSH launcher"
arch=('loong64' 'x86_64')
url="https://github.com/xvxx/shy"
license=('MIT')
depends=('gcc-libs')
makedepends=('rust')
source=("$pkgname-$pkgver.tar.gz::$url/archive/v$pkgver.tar.gz")
sha256sums=('8a3a2ec41ea648274e741a6fd326210b3c1a0bd8a3b1f7471980740569de48d0')

build() {
  cd "$pkgname-$pkgver"
  cargo build --release --locked
}

check() {
  cd "$pkgname-$pkgver"
  cargo test --release --locked
}

package() {
  cd "$pkgname-$pkgver"
  install -Dm 755 "target/release/$pkgname" -t "$pkgdir/usr/bin"
  install -Dm 644 README.md -t "$pkgdir/usr/share/doc/$pkgname"
  install -Dm 644 LICENSE -t "$pkgdir/usr/share/licenses/$pkgname"
  install -Dm 644 "doc/$pkgname.1" -t "$pkgdir/usr/share/man/man1"
}

# vim: ts=2 sw=2 et:
